
本日の内容・連絡事項

FreeFem++ のインストールは、適切に処理すれば出来るようなの
で、レポート課題 3.2は出さない (レポート課題 3.1だけとする)。
FreeFem++ のインストールがうまく行かない人は遠慮せず、相談し
て下さい。
本日の目標: レポート課題 3に必要なことを説明し切ること。

Dirichletの原理 — ポテンシャル問題が、弱定式化できたり、変分問
題に変換できる有名な事実の紹介
この後、より一般の Poisson 方程式の境界値問題に対する弱形式を導
出する (少し先の §4.7.2 の一部を説明する)。
有限要素法の概略 (お話)
サンプル・プログラム potential2d-v0.edp を (もう一度) 見る。
FreeFem++ の文法を説明する。

https://m-katsurada.sakura.ne.jp/lecture/applied-complex-function-2023/report3.pdf


4.5 Dirichletの原理
Laplace方程式の Dirichlet境界値問題

△u = 0 (in Ω),(1a)

u = g (on ∂Ω)(1b)

の解 uの存在を示すため、Riemann は次のように考えた。

境界条件 (1b) を満たす関数の全体 X と、X 上の汎関数 J を考える。
X :=

{
u
∣∣ u : Ω → R, (∀x ∈ ∂Ω) u(x) = g(x)

}
.

J[u] :=

∫ ∫
Ω

(
u2x + u2y

)
dx dy (u ∈ X ).

Dirichletの原理� �
J の最小値を与える u は △u = 0 (in Ω) を満たす。� �
したがって J の最小値を与える u は (1a), (1b) の解である。
Riemann (1826–1866)は、Dirichlet (1805–1859)先生の講義の中で Dirich-

let の原理を聴いたそうである。

4.5 Dirichletの原理
証明

v : Ω → R は、v = 0 (on ∂Ω) を満たす任意の関数とする。任意の t ∈ R に対
して u + tv ∈ X である。仮定より

f (t) := J[u + tv ] (t ∈ R)

は t = 0 で最小値をとる。

ところが
f (t) = J[u] + 2t

∫ ∫
Ω

(uxvx + uyvy )dx dy + t2
∫ ∫

Ω

(
v2
x + v2

y

)
dx dy

は t の 2次関数であり、t = 0 で最小となるので、1次の係数は 0 である:

(2)

∫ ∫
Ω

(uxvx + uyvy ) dx dy = 0.

Green の公式 (

∫ ∫
Ω

△uv dx dy =

∫
∂Ω

∂u

∂n
v dσ −

∫ ∫
Ω

∇u · ∇v dx dy) より∫ ∫
Ω

△u v dx dy = 0.

これが任意の v について成り立つことから (変分法の基本補題により)

△u = 0 (in Ω).

4.5 Dirichletの原理
反省

Riemann は、汎関数 J[u] を最小にする u ∈ X の存在は明らかだと考
えた。

J は下に有界 (J[u] ≥ 0) であるから、J は下限を持つ。それは最小値のはず…

それに Weierstrass が疑義を呈した (「下限は本当に最小値？」とツッ
コミを入れた)。これに Riemann は存命中に答えられなかった。
現代的な解説をすると、関数空間は無限次元空間なので、有界閉集合上

の連続関数であっても、最小値を持たないことがありえる。
ポテンシャル問題は重要なため、解の存在について、多くの人が努力し

て Dirichlet原理を用いない証明がいくつか発見されたが、Riemann の発
表から約 50年後 (1900年頃)、D. Hilbert が Dirichlet 原理に基づく証明を
発表し、肯定的に解決した。
今では解の存在証明は、このルートをたどるのがスタンダードになって

いる。…でも応用複素関数としては、ここから数値計算法に舵を切る (存
在証明については、関数解析か偏微分方程式論で学んでください)。

4.6 ポテンシャル問題の数値解法 (1) 有限要素法
ポテンシャル問題を数値的に解くことを考えよう。この「応用複素関数」で

は、有限要素法と基本解の方法を簡単に紹介する。

差分法で解くこともできるが、長方形領域でない問題を解くには工夫が必要に
なり、あまり便利でない。
有限要素法の主たるアイディアは次の 2つ:

(1) 弱形式を用いる。
(2) 領域を三角形、四面体などの有限要素に分割し、近似解や試験関数に区分
的多項式を採用する。

この講義では有限要素法の詳細は解説できないが、幸い FreeFem++ という
ソフトを用いると、弱形式さえ分かれば、有限要素についてはソフトに任せにし
て、数値計算ができる。
実は Dirichlet 原理の証明中に現れた (10) は Laplace方程式の Dirichlet 境界

値問題の弱形式である。(弱形式については、次回解説を行う。)

今回は「百聞は一見にしかず」で、 まずはプログラム (スライド 1枚)を紹介
する。

2,3行書き換えるだけで「自分の問題」が解ける。

は、有限要素法と基本解の方法を簡単に紹介する。
差分法で解くこともできるが、長方形領域でない問題を解くには工夫が必要に

なり、あまり便利でない。
有限要素法の主たるアイディアは次の 2つ:

(1) 弱形式を用いる。
(2) 領域を三角形、四面体などの有限要素に分割し、近似解や試験関数に区分
的多項式を採用する。

この講義では有限要素法の詳細は解説できないが、幸い FreeFem++ という
ソフトを用いると、弱形式さえ分かれば、有限要素についてはソフトに任せにし
て、数値計算ができる。
実は Dirichlet 原理の証明中に現れた (10) は Laplace方程式の Dirichlet 境界

値問題の弱形式である。(弱形式については、次回解説を行う。)

今回は「百聞は一見にしかず」で、 まずはプログラム (スライド 1枚)を紹介
する。

2,3行書き換えるだけで「自分の問題」が解ける。

// potential2d-v0.edp --- 2次元非圧縮ポテンシャル流
// 速度ポテンシャル，速度を求め、等ポテンシャル線, 速度場を描く
border Gamma(t=0,2*pi) { x = cos(t); y = sin(t); } // 円盤領域
int m=40;

mesh Th=buildmesh(Gamma(m));

plot(Th, wait=1, ps="Th.eps");

// 次の 2行は区分 1次多項式を使うという意味
fespace Vh(Th,P1);

Vh phi, v, v1, v2;

// 境界条件の設定
func Vn=x+2*y; // Ωが単位円で, V=(1,2) のとき V・n=x+2y

// 速度ポテンシャルφを求め、その等高線 (等ポテンシャル線) を描く
solve Laplace(phi,v) =

int2d(Th)(dx(phi)*dx(v)+dy(phi)*dy(v)) -int1d(Th,Gamma)(Vn*v);

plot(phi,ps="contourpotential.eps",wait=1);

// ベクトル場 (v1,v2)=∇φ を描く (ちょっと雑なやり方)

v1=dx(phi); v2=dy(phi);

plot([v1,v2],ps="vectorfield.eps",wait=1);

// 等ポテンシャル線とベクトル場を同時に描く
plot([v1,v2],phi,ps="both.eps", wait=1);

サンプル・プログラムを読む

前回の授業で紹介した potential2d-v0.edp をまだ入手していない人
は、次のようにダウンロードして、自分の使い慣れたテキスト・エディ
ターで開いてみよう (簡単な解説はレポート課題 3に書いてある)。� �

curl -O https://m-katsurada.sakura.ne.jp/complex2/potential2d-v0.edp� �
実行はターミナルで次のように行う。� �
FreeFem++ potential2d-v0.edp� �

enter キー (  ) で次のグラフィックスに進み、p で一つ前のグラフィッ
クに戻る。esc で終了する。

5 FreeFem++の文法
5.1 はじめに

FreeFem++ は有限要素法によって微分方程式の数値シミュレーションを行うためのソ
フトウェアであり、言語処理系である (Hecht [1])。
インタープリターである (その点は MATLABや Pythonと似ている)。

今回は、プログラミング言語としての FreeFem++を説明する (マニュアルを見ても良
く分からない — 少なくとも私は)。

FreeFem++のことを「有限要素法専用ツール」と考える人もいる。確かに有限要素法
に便利な命令が組み込まれているが、それ以外の目的のプログラミングに必要な機能も十
分に備わっている (実際、有限体積法や差分法のプログラムも記述可能である)。効率を度
外視すれば、Cのようなプログラミング言語で出来ることは FreeFem++でも出来る、と
考えよう。
文法は、C++に似ている (ゆえに Cにも似ている)。Cしか知らない人は、C++のスト

リームを使った入出力 (cout, cin の利用) を調べておくこと ([2] の付録に書いておいた)。
参考: FreeFem++ は C++ で記述されている。
マニュアル Hect[3] は事例集の性格が強く、言語仕様が整理された形では載っていな

い。以下の説明は、個人的なノートである桂田 [2] に基づく。
マニュアル以外の情報源として、日本応用数理学会のチュートリアル (鈴木 [4], [5])、

テキスト大塚・高石 [6] (本学学生は Maruzen eBook で読める) などがある。

5.1 はじめに 基本的な Poisson 方程式のプログラム
curl -O https://m-katsurada.sakura.ne.jp/program/fem/poisson.edp で入手できる。

// poisson.edp

// 境界の定義 (単位円), いわゆる正の向き
border Gamma(t=0,2*pi) { x=cos(t); y=sin(t); }

// 三角形要素分割を生成 (境界を 50 に分割)

mesh Th = buildmesh(Gamma(50));

plot(Th,wait=true); // plot(Th,wait=true,ps="Th.eps");

// 有限要素空間は P1 (区分的１次多項式) 要素
real [int] levels =-0.012:0.001:0.012;

fespace Vh(Th,P1);

Vh u,v;

// Poisson 方程式 -△ u=f の右辺
func f = x*y;

// 問題を解く
solve Poisson(u,v)

= int2d(Th)(dx(u)*dx(v)+dy(u)*dy(v))-int2d(Th)(f*v)

+on(Gamma,u=0);

// 可視化 (等高線)

plot(u,wait=true);

//plot(u,viso=levels,fill=true,wait=true);

// 可視化 (3 次元) --- マウスで使って動かせる
plot(u,dim=3,viso=levels,fill=true,wait=true);

→ 独特の命令ばかりで、汎用のプログラミング言語の機能があることは分かりにくい。かつらだ
桂 田

5.2 汎用のプログラミング機能 5.2.1 C言語と似ているところ

改めて数えるととても多い。

// から行末までは注釈、/* と */ で挟まれた部分は注釈 (共に C 言語と同じ)

文の最後は ;

四則演算 (+, -, *, /) や、代入 (=) などの演算子
0 は偽、0以外の整数は真とみなす。一方、比較演算・論理演算などの結果は 0
(false) または 1 (true).

変数宣言の文法も C言語と同様。型名の後に, で区切った名前のリストを書く。
関数呼び出しの文法も C言語と同様。
ブロックは { と } で複数 (0個以上)の文を囲んで作る。
比較演算子 (==, !=, <, <=, >, >=)、論理演算子 (&&, ||, !)、if, if else などの制
御構造。
ただし switch はない。
for, while などの繰り返し制御。break (ループを抜ける), continue (次の繰り返
し) など。
ただし do while はない。
数学関数の名前

他にもあるだろう…
FreeFem++ のソース・プログラムは C++ 言語で書かれているため、「C言語に似て

いる」は本当は「C++言語に似ている」かもしれない。

5.2.2 C++言語と似ているところ: cin, cout を用いた入出力

cin, cout を使ったストリーム・入出力は、C++ を知っている人には簡単であろう。
C++については、付録??(準備中　) を適宜参考にすること。

C で printf() で出力したものは、cout << で置き換えられる。� �
cout << "Hello world" << endl; // printf("Hello, world\n");

cout << "n=" << n << ", m=" << m << endl; // printf("n=%d, m=%d\n",n,m);

cout << "x=" << x << endl; // printf("x=%f\n", x);� �
書式指定については後述する。

C で scanf() で入力したものは、cin >> 変数名 で置き換えられる。� �
cin >> a >> b; // scanf("%d %d", &a, &b);

cin >> x; // scanf("%lf", &x);� �
外部ファイルとの入出力については、https://m-katsurada.sakura.ne.jp/labo/

text/freefem-note/node31.html#section:_____________________ を見よ。

かつらだ
桂 田

まさし
祐 史 応用複素関数 第 12 回 ～ ポテンシャル問題 (3) ～ 11 / 55

https://m-katsurada.sakura.ne.jp/labo/text/freefem-note/node31.html#section:_____________________
https://m-katsurada.sakura.ne.jp/labo/text/freefem-note/node31.html#section:_____________________


5.2.3 データ型
整数を表すための int がある (C言語の int に相当)

実数を表すための real がある (C言語の double に相当)

複素数を表すための complex がある (C言語の complex に相当, 実部・虚部が
double)
例えば complex a=1+2i; 入出力は 2次元ベクトル風の (1,2)

論理を表すための bool がある (C言語の bool に相当). true, false と言う値が
あるが、それぞれ 1, 0 の別名と考えて良い。
例えば plot(u,wait=true); は plot(u,wait=1); と同じ。
文字列を表すための string がある (C++言語の string に相当, 日本語不可？)。

2つの string s1, s2を、(+ 演算子を用いて) s1+s2 で連結できる。
string+数値 とすると、数値を文字列に変換してから連結する。� �
real a=1.23, b=4.56;

string s;

s= "a=" + a + ", b=" + b + ".";

cout << s << endl;� �
string を int に変換する atoi(), string を real に変換する atof() が
ある (C言語の真似)。

double)
例えば complex a=1+2i; 入出力は 2次元ベクトル風の (1,2)

論理を表すための bool がある (C言語の bool に相当). true, false と言う値が
あるが、それぞれ 1, 0 の別名と考えて良い。
例えば plot(u,wait=true); は plot(u,wait=1); と同じ。
文字列を表すための string がある (C++言語の string に相当, 日本語不可？)。

2つの string s1, s2を、(+ 演算子を用いて) s1+s2 で連結できる。

string+数値 とすると、数値を文字列に変換してから連結する。� �
real a=1.23, b=4.56;

string s;

s= "a=" + a + ", b=" + b + ".";

cout << s << endl;� �
string を int に変換する atoi(), string を real に変換する atof() が
ある (C言語の真似)。

かつらだ

5.2.4 配列型 1次元
1次元配列は、C言語に (少し)似ている。
配列 a の第 i 要素は、a[i] としてアクセスできるが、a(i) としてアク

セスするのが普通？� �
real[int] a1(3); // Cで double a1[3]; とするのに似ている
for (int i=0;i<3;i++)

a1(i)=i; // a1[i]=i; としても良い。

real[int] a2 = [0,1,2]; // C で double a[]={0,1,2}; とするのに似てる
real[int] a3 = 0:2; // これは少し MATLAB風

cout << "a1=" << a1 << endl;

cout << "a2=" << a2 << endl;

cout << "a3=" << a3 << endl;� �
追記: a1 の要素数は a1.n で得られる。

5.2.4 配列型 2次元
2次元配列は少し違う。2次元配列 a の (i , j) 要素にアクセスするには a(i , j) とする。� �
real[int,int] kuku(9,9);

int i,j;

for (i=0; i<kuku.n; i++) {

for (j=0; j<kuku.m; j++) {

kuku(i,j)=(i+1)*(j+1);

cout << setw(3) << kuku(i,j);

}

cout << endl;

}

cout << kuku << endl;

real[int,int] kuku2=[[1,2,3,4,5,6,7,8,9],

[2,4,6,8,10,12,14,16,18],

[3,6,9,12,15,18,21,24,27],

[4,8,12,16,20,24,28,32,36],

[5,10,15,20,25,30,35,40,45],

[6,12,18,24,30,36,42,48,54],

[7,14,21,28,35,42,49,56,63],

[8,16,24,32,40,48,56,64,72],

[9,18,27,36,45,54,63,72,81]];

cout << kuku2 << endl;� �かつらだ

5.2.5 FreeFem++の real データの入出力の書式指定
何も指定しないと C言語の %g 相当の出力になる。
cout.precision(n); とすると、以下小数点以下の桁数は nになる。
cout.precision(15);

cout << "pi=" << pi << endl;

幅を指定するには << setw(桁数) とする (これは毎回必要)。
cout << "pi=" << setw(20) << pi << endl;

cout.fixed; とすると、以下固定小数点数形式 (C 言語の %f相当)
になる。
cout.fixed;

cout << "NA=" << NA << endl;

cout.scientific; とすると、以下指数形式 (C言語の %e相当) に
なる。
cout.scientific;

cout << "pi=" << pi << endl;

cout.default; とすると、以下デフォールト (C言語の%g) に戻る。
かつらだ

5.2.5 FreeFem++の real データの入出力の書式指定 例
// testfloat.edp

real NA = 6.022e+23;

// デフォールト %g に相当
cout << "pi=" << pi << ", NA=" << NA << ", pi*NA=" << pi * NA << endl << endl;

// 幅を 20 に指定 %20g に相当
cout << "pi=" << setw(20) << pi << ", NA=" << setw(20) << NA

<< ", pi*NA=" << setw(20) << pi * NA << endl << endl;

// 小数点以下の桁数を 15 に指定 %20.15g に相当？
cout.precision(15);

cout << "pi=" << setw(20) << pi << ", NA=" << setw(20) << NA

<< ", pi*NA=" << setw(20) << pi * NA << endl << endl;

// 固定小数点数形式 %.15f に相当
cout.fixed;

cout << "pi=" << pi << ", NA=" << NA << ", pi*NA=" << pi * NA << endl << endl;

// %20.15f に相当
cout << "pi=" << setw(20) << pi << ", NA=" << setw(20) << NA

<< ", pi*NA=" << setw(20) << pi * NA << endl << endl;

// 指数形式 %20.15e に相当
cout.scientific;

cout << "pi=" << setw(20) << pi << ", NA=" << setw(20) << NA

<< ", pi*Na=" << setw(20) << pi * NA << endl << endl;

// %g 形式に戻す %.15g に相当
cout.default;

cout << "pi=" << pi << ", NA=" << NA << ", pi*Na=" << pi * NA << endl;かつらだ

5.3 有限要素法のための機能
5.3.1 有限要素法のプログラムの構成

有限要素法のプログラムと言っても色々あるが、基本的と考えられる 2
次元 Poisson方程式の境界値問題のプログラムを例にして説明する。

1 領域の定義と領域の三角形分割
2 有限要素空間の定義
3 弱形式を次のいずれかで定義して解く。

a solve()

弱形式を与えると同時にそれを解く (弱解を求める)。
b problem()

弱形式を与えて問題を解く関数を定義する。発展問題で便利。
c varf, matrix
弱形式を与えて連立 1次方程式を作る。

かつらだ

5.3.2 領域の定義と領域の三角形分割 (2次元の場合)

問題を考える領域を定義し、三角形分割をすることが必要である。

2次元 (有界)領域の多くは、その境界曲線を定義することで定まる。
境界曲線は border と言う型の変数として定義される。
三角形分割は mesh と言う型の変数として定義される。
buildmesh() と言う関数は、各 border を何等分するか指定すること
で、borderの囲む領域を三角形分割して、mesh型のデータを作る。
mesh 型のデータは、readmesh(), writemesh() と言う関数を用い
て入出力できる (フォーマットはテキスト・ファイル)。
mesh 型のデータは、plot() により可視化できる。
矩形領域 (辺が座標軸に平行な長方形) は、square() と言う命令で
mesh 型データが作れる (参考「FreeFem++ノート」)。

5.3.2 領域の定義と領域の三角形分割 border

円周全体を C とする� �
border C(t=0,2*pi) { x=cos(t); y=sin(t); }� �

円周の上半分、下半分を別々に Gamma1, Gamma2 と定義する� �
int C=1;

...

border Gamma1(t=0,pi) { x=cos(t); y=sin(t); label=C; }

border Gamma2(t=pi,2*pi) { x=cos(t); y=sin(t); label=C; }� �
正方形領域 (0, 1)× (0, 1) の 4つの辺 C1,C2,C3,C4を定義� �

border C1(t=0,1) { x=t; y=0; label=1; }

border C2(t=0,1) { x=1; y=t; label=2; }

border C3(t=0,1) { x=1-t; y=1; label=3; }

border C4(t=0,1) { x=0; y=1-t; label=4; }� �
(label の値指定は必須ではない。指定する場合は 0 以外の値を選ぶ。)

5.3.2 領域の定義と領域の三角形分割 メッシュ (mesh)

それぞれ表示してみる� �
// 例 1

border C(t=0,2*pi) { x=cos(t); y=sin(t); }

mesh Th1=buildmesh(C(50));

plot(Th1,wait=true,ps="Th1.eps");

// 例 2

int C0=1;

border Gamma1(t=0,pi) { x=cos(t); y=sin(t); label=C0; }

border Gamma2(t=pi,2*pi) { x=cos(t); y=sin(t); label=C0; }

mesh Th2=buildmesh(Gamma1(25)+Gamma2(50));

plot(Th2,wait=true,ps="Th2.eps");

// 例 3

border C1(t=0,1) { x=t; y=0; label=1; }

border C2(t=0,1) { x=1; y=t; label=2; }

border C3(t=0,1) { x=1-t; y=1; label=3; }

border C4(t=0,1) { x=0; y=1-t; label=4; }

mesh Th3=buildmesh(C1(20)+C2(20)+C3(20)+C4(20));

plot(Th3,wait=true,ps="Th3.eps");� �かつらだ

5.3.2 領域の定義と領域の三角形分割 メッシュ (mesh)

図 1: C(50) 図 2: Gamma1(25)+Gamma2(50) 図 3: C1(20)+C2(20)+...

5.3.2 領域の定義と領域の三角形分割 メッシュ (mesh)

有限個の Jordan閉曲線で囲まれた多重連結領域を三角形分割すること
もできる。

sampleMesh.edp� �
border a(t=0,2*pi){ x=cos(t); y=sin(t);label=1;}

border b(t=0,2*pi){ x=0.3+0.3*cos(t); y=0.3*sin(t);label=2;}

plot(a(50)+b(+30),wait=true,ps="border.eps");

mesh ThWithoutHole = buildmesh(a(50)+b(+30));

plot(ThWithoutHole,wait=1,ps="Thwithouthole.eps");

plot(a(50)+b(-30),wait=true,ps="borderminus.eps");

mesh ThWithHole = buildmesh(a(50)+b(-30));

5.3.2 領域の定義と領域の三角形分割 メッシュ (mesh)

普通は mesh データの細かいことは見る必要がないかもしれないが…

Th をメッシュとするとき、Th.nt は三角形の数 (the number of
triangles)、Th.nv は節点の数 (the number of vertices)、Th.area は
領域の面積 (area) である。
Th(i) は i 番目の節点 (i = 0, 1, . . . , Th.nv− 1)で、その座標は
Th(i).x と Th(i).y である。Th(i).label はその節点のラベル (領
域内部にあれば 0, それ以外は境界のどこか) を表す。
Th[i] は i 番目の三角形 (i = 0, 1, . . . , Th.nt− 1)、Th[i][j] は i 番
目の三角形の j 番目の節点 (j = 0, 1, 2) の全体節点番号、その節点
の座標は Th[i][j].x と Th[i][j].y である。三角形の面積は
Th[i].area である。
点 (x , y) を含む三角形の番号は Th(x,y).nuTriangle で得られる。

5.3.2 領域の定義と領域の三角形分割 メッシュ (mesh)

次のような場合に readmesh(), writemesh() は有効である。

(1) FreeFem++ を用いて三角形分割を行い、得られたメッシュ・デー
ターを外部のプログラムで利用する (有限要素解の計算は自作プログ
ラムで行う等)。

(2) 自作のプログラムで三角形分割を行い、そのメッシュ・データーを
FreeFem++ で利用する。

readmesh(), writemesh() で入出力されるメッシュ・データのフォー
マットについては、「FreeFem++ノート §6.2 mesh ファイルの構造」を
見よ。

Mesh 型の変数 Th の内容は、
writemesh(Th, "bunkatsu.msh");

のようにしてファイルに出力できる。
そのフォーマットにのっとって作られたファイルがあれば、

Mesh Th=readmesh("bunkatsu.msh");

のようにして読み込める。

5.3.3 有限要素空間
既に定義しておいた mesh 型データと、要素の種類を表す名前 (P1, P2, ...) を

用いて、有限要素空間 (この講義では X̃ のように表したが、Vh などの記号で表
すことが多い) を定義する。

fespace 型の変数は関数空間を表すことになる。
例えば Th と言う mesh型の変数があるとき、� �
fespace Vh(Th,P1);� �

とすると有限要素空間 Vhが定義される。
これは文法的には型名で� �
Vh u,v;� �

として変数 u, v が定義できる。これらが個々の関数を表す。
(数学語では u, v ∈ Vh という調子)

(注 これまでの授業で、三角形要素分割して、区分的 1次多項式 (P1要素)し
か紹介しなかったが、Poisson方程式の境界値問題以外では、他の要素 (P1b, P2,
P2Morley,...) が必要になることがある。)

5.3.3 有限要素空間

u の節点での値を集めた配列は u[] で表す。
u[].n (u.nでも同じ) は Th.nv と同じである。
i 番目の節点での値 (授業中の式で ui = û(Pi )) は u[](i)

u は補間多項式でもあり、(x , y) での値は u(x,y) で得られる。

5.3.3 弱形式を定義して解く
いよいよ弱形式を定義する方法の説明である。大きく分けて 3通りある。

(a) solve — 弱形式を与えると同時にそれを解く (弱解を求める)。
(b) problem — 弱形式を与えて問題を解く関数を定義する。
(c) varf, matrix — 弱形式を与えて連立 1 次方程式を作る。

これまで説明して来た次の Poisson 方程式の境界値問題を元に説明する。
−△u(x , y) = f (x , y) ((x , y) ∈ Ω)(3a)

u(x , y) = g1(x , y) ((x , y) ∈ Γ1)(3b)

∂u

∂n
(x , y) = g2(x , y) ((x , y) ∈ Γ2).(3c)

弱解 u とは、Xg1 に属し、次の弱形式を満たすものである。
(4) ⟨u, v⟩ = (f , v) + [g2, v ] (v ∈ X ).

ただし
(5) Xg1 := {w | w = g1 on Γ1} , X := {v | v = 0 on Γ1} .

5.3.3 弱形式を定義して解く (a) solve を利用
Poisson 方程式の境界値問題を解くサンプル・プログラム poisson.edp では、(a) を

用いた。
solve で弱形式を定義して解く� �

solve Poisson(u,v)=

int2d(Th)(dx(u)*dx(v)+dy(u)*dy(v))-int2d(Th)(f*v)-int1d(Th,2,3)(g2*v)

+on(1,4,u=g1);� �

次はどの方法でも共通である。
dx(), dy() はそれぞれ x , y での微分を表す。
高階の微分は dxx(), dxy(), dyy() のようにする。
int2d(Th) は、Th の領域全体の積分 (重積分)を表す。
また int1d(Th,2,3) は境界のうち、ラベルが 2,3である部分 (正方形の右と上)の
積分 (境界積分、今の場合は線積分)を表す。
+on(1,4,u=g1) は境界のうち、ラベルが 1,4である部分 (正方形の下と左)で、
u = g1 という Dirichlet 境界条件を課すことを表す (+on(1,u=g1)+on(4,u=g1) と
分けて書くことも可能)。
ベクトル値関数の場合は、+on(1,u1=g1,u2=g2) のように複数の方程式を書くこと
もできる。

以下、この問題の場合に、(b), (c) がどうなるか示す。

かつらだ
桂 田

まさし
祐 史 応用複素関数 第 12 回 ～ ポテンシャル問題 (3) ～ 28 / 55

https://m-katsurada.sakura.ne.jp/program/fem/poisson.edp

5.3.3 弱形式を定義して解く (a) solve を利用
Poisson 方程式の境界値問題を解くサンプル・プログラム poisson.edp では、(a) を

積分 (境界積分、今の場合は線積分)を表す。

5.3.3 弱形式を定義して解く (b) problem を利用
(b) problem を利用する方法では、次のようになる。

problem で弱形式を定義して解く� �
problem Poisson(u,v)=

int2d(Th)(dx(u)*dx(v)+dy(u)*dy(v))-int2d(Th)(f*v)-int1d(Th,2,3)(g2*v)

+on(1,4,u=g1);

Poisson;� �
この問題の場合は、solve と比べての利点は特に感じられないかもしれない

が、時間発展の問題では、同じ形の弱形式を何度も解く必要が生じるので、有効
である (効率が上がる可能性がある — 後述)。

5.3.3 弱形式を定義して解く (c) varf, matrix を利用
varf, matrix を利用� �

real Tgv=1.0e+30; // tgv と小文字でも可
varf a(u,v)=

int2d(Th)(dx(u)*dx(v)+dy(u)*dy(v))

+on(1,4,u=g1);

matrix A=a(Vh,Vh,tgv=Tgv,solver=CG);

varf l(unused,v)=

int2d(Th)(f*v)+int1d(Th,2,3)(g2*v)

+on(1,4,unused=0);

Vh F;

F[]=l(0,Vh,tgv=Tgv);

u[]=A^-1*F[];� �
あらすじは、連立 1次方程式 Au = f の A, f を別々に計算して、A−1f を計算するこ

とで u を得る、ということである (詳細は、実は現時点で把握していないので省略する)。

tgv (terrible great value) は以前説明した (tgv = 1030)。
solver= は連立 1次方程式の解法を指定する。

CG CG法 (共役勾配法) (反復法, 正値対称行列 (spd) 用)
GMRES GMRES法 (反復法, 一般の正則行列用)
UMFPACK UMFPACKを利用 (直接法, 一般の正則行列用)
sparsesolver ダイナミック・リンクで外部のソルバーを呼ぶ

以上で 5節を終わり、4節「ポテンシャル問題」に戻ります。
(2023/7/11日の授業では、弱形式 (10) の導出だけ行いました。その他

も説明したいところですが、今年度はあきらめます。)

4.7 弱解の方法 4.7.1 はじめに

今回用いる有限要素法は、今では微分方程式論で常識となっている弱解
の方法 (weak formulation) を応用したものである。
これは、Riemann による、Laplace 方程式の Dirichlet 境界値問題を解

くために用いた論法を発展させたものである。現在では様々な微分方程式
に応用されているが、ここではもっとも基本的な Poisson 方程式の境界値
問題について説明する (おおむね菊地 [7] に沿った解説)。

厳密に議論するには、広義導関数、いわゆる
ソ ボ レ フ
Sobolev 空間を導入する

必要がある。具体的には、後で出て来る Xg1 と X は、本当は Sobolev 空
間の一種 H1(Ω) を用いて

Xg1 :=
{
w ∈ H1(Ω)

∣∣ w = g1 on Γ1
}
, X :=

{
w ∈ H1(Ω)

∣∣ w = 0 on Γ1
}

と定義すべきものである。ともあれ、ここでは関数の滑らかさに関する議
論には目をつぶって議論する。

(Sobolev空間を学ぶときは、Brezis [8], [9] をチェックしてみよう。)

4.7.2 Poisson方程式の境界値問題 (P)

Laplace 方程式を一般化した Poisson方程式の境界値問題を考える。
Ω は Rm (m = 2, 3) の有界領域、Γ := ∂Ω は Ω の境界、Γ1 と Γ2 は

Γ1 ∪ Γ2 = Γ, Γ1 ∩ Γ2 = ∅

を満たす。また、f : Ω → R, g1 : Γ1 → R, g2 : Γ2 → R が与えられている
とする。Γ1 = ∅ (全周 Neumann) のときは

∫
Γ2

g2 dσ = 0 を仮定する。

問題 (P)� �
Find u s.t.

−△u = f (in Ω),(6)

u = g1 (on Γ1),(7)

∂u

∂n
= g2 (on Γ2).(8)� �かつらだ

4.7.3 弱定式化 (W) と変分問題 (V)

Xg1 :=
{
w

∣∣ w : Ω → R, w = g1 on Γ1

}
, X :=

{
w

∣∣ w : Ω → R, w = 0 on Γ1

}
,

(9) J[w ] :=
1

2

∫
Ω

|∇w |2 dx −
∫
Ω

f w dx −
∫
Γ2

g2 w dσ (w ∈ Xg1).

とおく。X の要素はしばしば試験関数 (test function) と呼ばれる。
次の 2つの問題を考える。

(W)� �
Find u ∈ Xg1 s.t.

(10)

∫
Ω

∇u · ∇v dx =

∫
Ω

f v dx +

∫
Γ2

g2 v dσ (v ∈ X ).

(条件 (10) を弱形式 (weak form) と呼ぶ。)� �
(V)� �

Find u ∈ Xg1 s.t.

(11) J[u] = inf
w∈Xg1

J[w ]. (inf は結局は min)� �

とおく。X の要素はしばしば試験関数 (test function) と呼ばれる。
次の 2つの問題を考える。

(W)� �
Find u ∈ Xg1 s.t.

(10)

∫
Ω

∇u · ∇v dx =

∫
Ω

f v dx +

∫
Γ2

g2 v dσ (v ∈ X ).

(条件 (10) を弱形式 (weak form) と呼ぶ。)� �
(V)� �

Find u ∈ Xg1 s.t.

(11) J[u] = inf
w∈Xg1

J[w ]. (inf は結局は min)� �かつらだ

4.7.4 3つの問題 (P), (W), (V) の同等性
(P), (W), (V) は、ほぼ同値な問題である。

実際、次が成り立つ。
定理 12.1
(1) u が (P) の解 ⇒ u が (W) の解
(2) u が (W) の解 ⇔ u が (V) の解
(3) u が (W) の解かつ u が C 2 級 ⇒ u が (P) の解

(授業では (2) の証明を省略した。そのため、次の補題も省略した。)
(2) の証明のために補題を準備する (証明は単なる計算であるので省略する)。

補題 12.2

任意の w ∈ Xg1 , v ∈ X , t ∈ R に対して次式が成立する。

J[w + tv ] =
t2

2

∫
Ω

|∇v |2 dx(12)

+ t

(∫
Ω

∇w · ∇v dx −
∫
Ω

f v dx −
∫
Γ2

g2 v dσ

)
+ J[w ].

4.7.5 (P)の解は (W)の解 — 弱形式の導出
(1)の証明
u が (P) の解と仮定する。(6) −△u = f に任意の v ∈ X をかけて Ω 上で積分すると

(13) −
∫
Ω

△u v dx =

∫
Ω

fv dx .

左辺を Greenの公式 ([10]) を用いて変形してから、v = 0 (on Γ1) と (8) ∂u
∂n = g2 を

用いると

−
∫
Ω

△u v dx = −
∫
Γ

∂u

∂n
v dσ +

∫
Ω

∇u · ∇v dx

= −
∫
Γ1

∂u

∂n
v dσ −

∫
Γ2

∂u

∂n
v dσ +

∫
Ω

∇u · ∇v dx

= −
∫
Γ2

g2 v dσ +

∫
Ω

∇u · ∇v dx .

(13) に代入して移項すると∫
Ω

∇u · ∇v dx =

∫
Ω

f v dx +

∫
Γ2

g2 v dσ.

すなわち、u は弱形式を満たす ((W) の解である)。

4.7.6 (W) と (V) は同値
(2)の証明
[u が (V) の解 ⇒ u は (W) の解] (デジャブ？ Dirichlet原理の証明と似ている)

u は (V) の解とする。任意の v ∈ X に対して、f (t) := J[u+ tv ] (t ∈ R) とおくと、
f (t) = J[u + tv ] ≥ J[u] = f (0).

ゆえに f は t = 0 で最小になる。ゆえに J[u+ tv ] の t の 1次の項の係数は 0 である:∫
Ω

∇u · ∇v dx −
∫
Ω

f v dx −
∫
Γ2

g2 v dσ = 0.

ゆえに u は (W) の解である。

[u が (W) の解 ⇒ u は (V) の解]

u は (W) の解とする。任意の w ∈ Xg1 に対して、v := u−w とおくと v ∈ X である。
J[w ]− J[u] = J[u + 1 · v ]− J[u] (t = 1 として補題 12.2を適用)

=
1

2

∫
Ω

|∇v |2 dx +

(∫
Ω

∇u · ∇v dx −
∫
Ω

f v dx −
∫
Γ2

g2 v dσ

)
=

1

2

∫
Ω

|∇v |2 dx ≥ 0.

ゆえに J[u] は J[w ] の最小値である。すなわち u は (V) の解である。

4.7.7 (W) の解が滑らかならば (P) の解
(3)の証明 u が (W) の解であり、かつ十分滑らかと仮定する。(W) の解であるから、∫

Ω

∇u · ∇v dx =

∫
Ω

f v dx +

∫
Γ2

g v dσ (v ∈ X )

を満たす。

さらに滑らかであるので、左辺に Green の公式が適用できて

(⋆)

∫
Γ2

∂u

∂n
v dσ −

∫
Ω

△u v dx =

∫
Ω

f v dx +

∫
Γ2

g2 v dσ (v ∈ X ).

特に v ∈ C∞
0 (Ω) の場合を考えると (境界上の積分が 0 なので)

−
∫
Ω

△u v dx =

∫
Ω

f v dx (v ∈ C∞
0 (Ω)).

変分法の基本補題より
−△u = f (in Ω).

これを (⋆) に代入すると ∫
Γ2

∂u

∂n
v dσ =

∫
Γ2

g2 v dσ (v ∈ X ).

再び変分法の基本補題より、∂u

∂n
= g2 (on Γ2). ゆえに u は (P) の解である。 証明終

……… 以上、Dirichlet原理の一般化

4.7.8 補足 変分法の基本補題
「任意の」(実際には「何かの条件を満たすすべての」) 関数 φ について∫

Ω

f (x)φ(x) dx = 0 が成り立つならば、f = 0 (in Ω), という形の命題を変分
法の基本補題 (fundamental lemma of calculs of variations) という。
色々なバージョンがあるが、次の形のもので用が足りることが多い。

命題 12.3 (変分法の基本補題)

Ω は Rn の開集合、f : Ω → C は局所可積分関数で

(∀φ ∈ C∞
0 (Ω))

∫
Ω

f (x)φ(x) dx = 0

を満たすならば
f = 0 (a.e. on Ω).

Ω で局所可積分とは、Ω に含まれる任意のコンパクト集合上で Lebesgue 積分可能と
いうこと。

f = 0 (a.e on Ω) とは、Ω に含まれるある測度 0の集合 N を除いて f = 0 ということ。
C∞
0 (Ω) については、次のページで説明する。

4.7.8 補足 C∞
0 (Ω)

C∞
0 (Ω) という記号は、解析学で頻出する (知っておくと良い)。

Ω を Rn の開集合とするとき、C∞
0 (Ω) を

C∞
0 (Ω) :=

{
v ∈ C∞(Ω)

∣∣ supp v はコンパクトで、Ω に含まれる}
で定める。ここで supp v は v の台 (the support of v) と呼ばれる集合で、
次式で定められる。

supp v := {x ∈ Ω | v(x) ̸= 0}.

ここで は、Rn における閉包を意味する。
コンパクトとは、Rn の部分集合 K については、K が有界な閉集合と

いうことである。
v ∈ C∞

0 (Ω) とは、v が C∞ 級で、∂Ω のある近傍では v = 0 を満たす
ことを意味する。

4.7.9 定理の使い道
(P) の解を求めたり、一意的な存在を示したいわけであるが、代わりに

(W) (あるいは (V)) を考える。

(W) の解の一意的な存在を示すのは比較的容易である。また (W) の近
似解を求めるのも簡単である (有限要素法がまさにそれをしてくれる)。

(W) の解が本当に (P) の解であるか？が問題になる。言い換えると
(W) の解 u は滑らかだろうか？

この問は、一見細かいことのようだが、実はとても重要である。
良く知られた (部分的な) 解答

Ω が C 2 級であれば (どういう意味？) Yes.

Ω が多角形の場合、凸ならば Yes, 凸でないならば一般には No.

(FreeFem++ の例で、L 字型の領域や、立方体から小さい立方体を除い
た領域がしばしば登場するが、このあたりのこと (領域の凸性) を問題に
しているわけである。)

2022/6/20 の授業は、4.7 の説明までで終わりました。
次回以降に、以下の 4.8 を講義するかどうかは、まだ決めていません。

4.8 ポテンシャル問題の数値解法 (2) 基本解の方法
4.8.1 −△ の基本解
次の関数 E は、−△ の基本解 (fundamental solution)と呼ばれる。

(14) E (x) :=


− 1

2π
log |x | (2次元の場合, x ∈ R2 \ {0})

1

4π

1

|x |
(3次元の場合, x ∈ R3 \ {0}).

一体何者か？
数学的な解答 次を満たす。ここで δ は Dirac のデルタ超関数。

(15) −△E = δ (△ def.
=

n∑
j=1

∂2

∂x2j
).

物理的な解答 (解釈) E は原点に置かれた単位点電荷の作る電場のポ
テンシャル (電位) である。

(Cf. 密度 ρで分布する場合のポテンシャル u は −△u = ρ を満たす。)

4.8.1 −△ の基本解
なぜ基本解は重要か？

重ね合わせることで “任意” の電荷分布 ρ のポテ
ンシャルが得られる。
定理 (のようなもの) Poisson方程式の特解

Ω ⊂ Rn, ρ : Ω → R が与えられたとき
(16) u(x) :=

∫
Ω
E (x − y)ρ(y) dy (x ∈ Ω)

とおくと次式が成り立つ。

(17) −△u = ρ (in Ω).

E には特異性があるので、(17) を証明するのは少し難しい (神保 [11]など)。
物理的には次のように納得できる。
微小体積 dy に存在する電荷は ρ(y) dy で、それが作る電場のポテンシャルは (基本解

を平行移動したものの電荷量倍で) E(x − y)ρ(y) dy . それを Ω 全体でトータルした u が
ポテンシャルになる。実際、E := − grad u は電場で、Maxwell の方程式の 1つ div E = ρ

から、− div grad u = ρ が得られる。すなわち (17) が成り立つ。

4.8.2 基本解の方法のアルゴリズム (電荷の作る電場の電位でポテンシャルを近似)

ポテンシャル問題の数値解法 (近似解法) への応用「基本解の方法 (the
method of fundamental solutions)」を紹介する。

Dirichlet 境界値問題を考えよう (Neumann境界値問題でも同様)。

△u = 0 (in Ω)(18)

u = g (on ∂Ω).(19)

ここで Ω は Rn (n = 2 or n = 3) の領域である。

Ω の外部に、Ω を取り囲むように、有限個の点 y1, · · · , yN を取り、各
yk に電荷量 Qk の電荷を置く。

4.8.2 アルゴリズム (電荷の作る電場の電位でポテンシャルを近似)

それらの電荷の作る電場のポテンシャルは

(20) u(N)(x) :=
N∑

k=1

QkE (x − yk).

△E = 0 (in Rn \ {0}) であるから、Qk の取り方によらず
△u(N)(x) = 0 (x ∈ Rn \ {y1, · · · , yN}). 特に △u(N) = 0 (in Ω).

後は Qk をうまく選んで、境界条件 (19) u = g (on ∂Ω) を近似的に満
たすようにする。
一つのやり方として、∂Ω 上に N 個の点 x1, · · · , xN を取って

(21) u(N)(xj) = g(xj) (j = 1, · · · ,N).

これで Qk (k = 1, · · · ,N) が定まることはすぐ分かる (次のスライド)。
非常に素朴な感じがするが、とてもうまく行くことが多い。

4.8.2 アルゴリズム (電荷の作る電場の電位でポテンシャルを近似)

(21) は次の連立 1次方程式と同値である。
E (x1 − y1) E (x1 − y2) · · · E (x1 − yN)
E (x2 − y1) E (x2 − y2) E (x2 − yN)

...
. . .

...
E (xN − y1) E (xN − y2) · · · E (xN − yN)



Q1

Q2
...

QN

 =


g(x1)
g(x2)

...
g(xN)

 .

Gaussの消去法などを用いて、Qk (k = 1, · · · ,N) が求められる。

(いわゆる密行列であるが、それほど大きな N は必要ないので、難しく
ない。)

4.8.3 基本解の方法の特徴

(1) ある ρ (0 < ρ < 1), C が存在して∥∥∥u − u(N)
∥∥∥ ≤ CρN (∥ · ∥ は適当なノルム)

が成り立つ (誤差の指数関数的減少, 次のスライドで数値例を示す)。しば
しば、高精度の解が非常に少ない計算量で得られることが期待できる。
Cf. 差分法, 有限要素法では、典型的な場合に

∥∥∥u − u(N)
∥∥∥ ≤ C

N2 .

(2) u(N) は調和関数である。特に grad u(N) の計算が簡単:

grad u(N)(x) = − 1

2π

N∑
k=1

Qk
x − yk

|x − yk |2
(2次元の場合).

(例えばポテンシャル流の計算を思い浮かべると、超便利と分かる。)

しかも
∥∥grad u − grad u(N)

∥∥ も指数関数的に減少する。
Cf. 差分法や有限要素法では、微分が難しかったり、精度が下がったりする。

(3) 理論的な基礎づけは、差分法、有限要素法と比べて不十分である。
(4) 同次方程式にしか適用できない、具体的な基本解が必要 → 汎用性は低い。
汎用性低いが、使えるときは、差分法・有限要素法に性能で勝る場合が多い。

汎用性低いが、使えるときは、差分法・有限要素法に性能で勝る場合が多い。かつらだ

4.8.3 基本解の方法の特徴 数値例
Ω が円盤 {

x ∈ R2
∣∣ |x | < 1

} の場合に、原点中心半径 R = 2 の円周上に一様
に電荷点 yk を配置した場合の近似解の精度を示す。
左が g(x) = Re [(x + iy)m], 右が g(x) = log |x − p| (p = (p, 0)) の場合。

図 4: m = 1, · · · , 5

図 5: p = 1.2, 1.4, · · · , 2.6. pが大き
い (特異点が遠い)ほど速く減衰

誤差の減少は非常に速い！

片側対数目盛で、直線上にのることから、誤差が指数関数的に減少している。
減少の速さについては研究されていて、ある程度まで説明可能である。

4.8.4 数値等角写像に対する天野の方法
天野要は、§4.4で述べた等角写像の求め方 (定理 8.6)と、基本解の方法

を組み合わせた、数値等角写像 (領域の写像関数を数値的に求めること)
の効率的なアルゴリズムを提唱した (天野 [12])。それを解説する。

§4.4で導入した記号を用いる。
u の近似 u(N) を基本解の方法で求めよう。N ∈ N に対して、{ζk}Nk=1

を「Ω を取り囲むように」 C \ Ω から選び、

(22) u(N)(z) :=
N∑

k=1

Qk log |z − ζk |

とおく。ここで Qk (k = 1, . . . ,N) は未知の実定数である。
{zj}Nj=1 を ∂Ω から選び、連立 1次方程式

(23) u(N)(zj) = − log |zj − z0| (j = 1, . . . ,N)

を解いて Qk (k = 1, . . . ,N) が求められる。
かつらだ

4.8.4 数値等角写像に対する天野の方法
u(N) の共役調和関数 v (N) を求めたい (u(N) を実部に持つ正則関数を求

めたい)。

天下りになるが、

(24) f (N)(z) := Q0 +
N∑

k=1

Qk Log
z − ζk
z0 − ζk

, Q0 :=
N∑

k=1

Qk log |z0 − ζk |

とおく。ここで Log は主値を表すとする (C \ (−∞, 0] を定義域とする)。

Re f (N)(z) =
N∑

k=1

Qk log |z0 − ζk |+
N∑

k=1

Qk log

∣∣∣∣ z − ζk
z0 − ζk

∣∣∣∣ = N∑
k=1

Qk log |z − ζk | = u(N)(z)

である。さらに

f (N)(z0) = Q0 +
N∑

k=1

Qk Log
z0 − ζk
z0 − ζk

= Q0 +
N∑

k=1

0 = Q0 ∈ R.

言い換えると Im f (N)(z0) = 0. この f (N) は、f = u + iv の良い近似であ
ると考えられる。

4.8.4 数値等角写像に対する天野の方法
以上をまとめると、次のアルゴリズムが得られる。

u(N)(z) :=
N∑

k=1

Qk log |z − ζk | ,(再掲 22)

u(N)(zj) = − log |zj − z0| (j = 1, . . . ,N),(再掲 23)

f (N)(z) := Q0 +
N∑

k=1

Qk Log
z − ζk
z0 − ζk

, Q0 :=
N∑

k=1

Qk log |z0 − ζk |(再掲 24)

天野のアルゴリズム� �
(0) {ζk}Nk=1 ⊂ C \ Ω, {zj}Nj=1 ⊂ ∂Ω を適当に選ぶ。
(1) (22), (23) で {Qk} を求める。
(2) (24) で f (N) を定める。
(3) φ(N)(z) := (z − z0) exp f

(N)(z) で定義される φ(N) を、等角写像
φ : Ω → D1 の近似として採用する。� �かつらだ

4.8.5 Jordan 領域の等角写像の計算プログラム
以下の Python プログラム conformalmap-v2.py では

Ω = D1
def.
= {z ∈ C | |z | < 1} , z0 =

1

2

の場合の Ω の写像関数、すなわち双正則な φ : Ω → D1 で
φ(z0) = 0, φ(z0) > 0

を満たすものを求める。この場合、実は次の 1次分数変換が解である。
φ(z) =

z − z0
1− z0z

.

プログラム入手 — ターミナルで次を実行� �
curl -O http://nalab.mind.meiji.ac.jp/~mk/complex2/conformalmap-v2.py� �

ターミナルで次のように実行� �
python conformalmap-v2.py� �かつらだ

4.8.5 Jordan 領域の等角写像の計算プログラム

図 6: z0 = 0.6 の場合. w = φ(z) による z 平面の原点中心の同心円、単位円の
半径の像を描いた。

